

graph twoway qfit — Twoway quadratic prediction plots[Description](#)
[Options](#)[Quick start](#)
[Remarks and examples](#)[Menu](#)
[Also see](#)[Syntax](#)

Description

`twoway qfit` calculates the prediction for *yvar* from a linear regression of *yvar* on *xvar* and *xvar*² and plots the resulting curve.

Quick start

Graph showing the quadratic prediction of *y* using *x* and *x*²

```
twoway qfit y x
```

Overlay quadratic prediction graph on a scatterplot of the observed data

```
twoway scatter y x || qfit y x
```

Same as above, but with a separate graph area for each level of *catvar*

```
twoway scatter y x || qfit y x, by(catvar)
```

Same as above, but with an additional graph area for all levels of *catvar* combined

```
twoway scatter y x || qfit y x, by(catvar, total)
```

Same as above, but with all graphs in a single row

```
twoway scatter y x || qfit y x, by(catvar, total row(1))
```

Menu

Graphics > Twoway graph (scatter, line, etc.)

Syntax

```
twoway qfit yvar xvar [if] [in] [weight] [, options]
```

<i>options</i>	Description
range (# #)	range over which predictions calculated
n (#)	number of prediction points
atobs	calculate predictions at <i>xvar</i>
estopts (<i>regress_options</i>)	options for regress
predopts (<i>predict_options</i>)	options for predict
<i>cline_options</i>	change look of predicted line
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All options are *rightmost*; see [G-4] **Concept: repeated options**.

yvar and *xvar* may contain time-series operators; see [U] **11.4.4 Time-series varlists**.

aweights, **fweight**s, and **pweight**s are allowed. Weights, if specified, affect estimation but not how the weighted results are plotted. See [U] **11.1.6 weight**.

Options

range(# #) specifies the *x* range over which predictions are calculated. The default is **range**(. .), meaning the minimum and maximum values of *xvar*. **range**(0 10) would make the range 0 to 10, **range**(. 10) would make the range the minimum to 10, and **range**(0 .) would make the range 0 to the maximum.

n(#) specifies the number of points at which predictions over **range**() are to be calculated. The default is **n**(100).

atobs is an alternative to **n**(). It specifies that the predictions be calculated at the *xvar* values. **atobs** is the default if **predopts**() is specified and any statistic other than **xb** is requested.

estopts(*regress_options*) specifies options to be passed along to **regress** to estimate the linear regression from which the curve will be predicted; see [R] **regress**. If this option is specified, commonly specified is **estopts**(nocons).

predopts(*predict_options*) specifies options to be passed along to **predict** to obtain the predictions after estimation by **regress**; see [R] **regress postestimation**.

cline_options specify how the prediction line is rendered; see [G-3] *cline_options*.

axis_choice_options associate the plot with a particular *y* or *x* axis on the graph; see [G-3] *axis_choice_options*.

twoway_options are a set of common options supported by all **twoway** graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over **by**() groups, and change some advanced settings. See [G-3] *twoway_options*.

Remarks and examples


Remarks are presented under the following headings:

Typical use
Cautions
Use with by()

Typical use

`twoway qfit` is nearly always used in conjunction with other `twoway` plottypes, such as

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. scatter mpg weight || qfit mpg weight
```



Results are visually the same as typing

```
. generate tempvar = weight^2
. regress mpg weight tempvar
. predict fitted
. scatter mpg weight || line fitted weight
```

Cautions

Do not use `twoway qfit` when specifying the `axis_scale_options` `yscale(log)` or `xscale(log)` to create log scales. Typing


```
. scatter mpg weight, xscale(log) || qfit mpg weight
```

produces something that is not a parabola because the regression estimated for the prediction was for `mpg` on `weight` and `weight^2`, not `mpg` on `log(weight)` and `log(weight)^2`.

Use with by()

qfit may be used with by() (as can all the twoway plot commands):

```
. scatter mpg weight || qfit mpg weight ||, by(foreign, total row(1))
```



Graphs by Car origin

Also see

[G-2] **graph twoway fpfit** — Twoway fractional-polynomial prediction plots

[G-2] **graph twoway lfit** — Twoway linear prediction plots

[G-2] **graph twoway line** — Twoway line plots⁺

[G-2] **graph twoway mband** — Twoway median-band plots

[G-2] **graph twoway mspline** — Twoway median-spline plots

[G-2] **graph twoway qfitci** — Twoway quadratic prediction plots with CIs

[R] **regress** — Linear regression

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).